






	<b>STMIK WIDYA CIPTA DHARMA SAMARINDA</b>	<b>S1 – TEKNIK INFORMATIKA</b>
<b>PRAK. TEKNIK MULTIMEDIA</b>	<b>GAME DRAG AND DROP ADOBE FLASH</b>	<b>Labsheet 02</b>
<i>Semester 4</i>		Dosen : Muhammad Ibnu Sa'ad, S.Kom., M.Kom Email : saad@wicida.ac.id



1. Buatlah symbol baru dengan nama **scoreMC** dan type **movie clip** kemudian check list **Export for ActionScript**. Kemudian Import gambar pada asset yang sudah disediakan di welearn. Lihat gambit berikut:





2. Selanjutnya buatlah symbol baru dengan nama **bintangMC**, langkahnya sama seperti membuat symbol **scoreMC**. Pada symbol **bintangMC** import gambar secara berurutan dengan memilih **Yes**. Lihat gambar berikut:




3. Langkah selanjutnya kita akan mengkombinasikan symbol **scoreMC** dan **bintangMC**. Caranya masuk ke symbol **scoreMC** kemudian drag and drop symbol **bintangMC** pada tools library bagian kanan, dan pada **instance name** buatlah nama **bintang**. Lihat gambar berikut:




4. Masih pada bagian symbol **scoreMC**. Buatlah text angka 0 dan pada instance name buatlah nama **nilaiBenar**. Klik button **Embed** dan checklist **Numerals**. Lihat gambar berikut:



5. Lakukan cara yang sama seperti langkah 4 untuk membuat **nilaiSalah**.





6. Kembali ke scan utama, caranya klik **Scene 1** pada bagian kiri atas atau gunakan shortcut **Ctrl+E**.



7. Langkah terakhir ubah Script pada layer action menjadi seperti coding berikut ini:

```
import flash.events.MouseEvent;

var jumlahSampah:int = 10;
var nilaiBenar:int = 0;
var sampahMasuk:int = 0;
var papanScore:scoreMC;

function setup():void{
    nilaiBenar = 0;
    sampahMasuk = 0;
```

```

//membuat sampah dengan perintah berulang
for (var i:int = 0; i < jumlahSampah; i++){
    var sampahMC:sampah = new sampah;
    sampahMC.x = 50+Math.random()*400;
    sampahMC.y = 50+Math.random()*300;
    sampahMC.awalX = sampahMC.x;
    sampahMC.awalY = sampahMC.y;
    sampahMC.gotoAndStop(Math.floor(Math.random()*10));
    //memberikan perintah kepada MC sampah dengan mouse
    sampahMC.addEventListener(MouseEvent.MOUSE_DOWN,
tekanMouse);
    sampahMC.addEventListener(MouseEvent.MOUSE_UP,
lepasMouse);
    addChild(sampahMC);
}
}

function tekanMouse(e:MouseEvent):void{
    var ob:Object = e.currentTarget;
    ob.startDrag();
}
function lepasMouse(e:MouseEvent):void{
    var ob:Object = e.currentTarget;
    ob.stopDrag();
    //menghitung jarak dengan movieclip tong sampah
    var jarak1:int = Math.sqrt((ob.x - tong1.x)*(ob.x - tong1.x)+(ob.y -
tong1.y)*(ob.y - tong1.y));
    var jarak2:int = Math.sqrt((ob.x - tong2.x)*(ob.x - tong2.x)+(ob.y -
tong2.y)*(ob.y - tong2.y));
    if (jarak1 < 100 || jarak2 < 100){
        //menentukan nilai berdasarkan frame (1-5 non organik, 6 - 10 organik)
        if ((ob.currentFrame < 6 && jarak1 < 100) || (ob.currentFrame > 5 &&
jarak2 < 100)){
            //memasukkan ke tempat yang benar
            nilaiBenar++;
        }
        sampahMasuk++;
        if (sampahMasuk >= jumlahSampah) tampilkanScore();
        //hapus objek
        ob.removeEventListener(MouseEvent.MOUSE_DOWN, tekanMouse);
        ob.removeEventListener(MouseEvent.MOUSE_UP, lepasMouse);
        removeChild(DisplayObject(ob));
    }else{
        //jika jarak lebih dari 100 sampah kembali ke posisi semula
        ob.x = ob.awalX;
        ob.y = ob.awalY;
    }
}

function tampilkanScore(){

```

```

papanScore = new scoreMC;
papanScore.x = 400;
papanScore.y = 240;
papanScore.nilaiBenar.text = String(nilaiBenar);
papanScore.nilaiSalah.text = String(jumlahSampah-nilaiBenar);
//menentukan bintang
if (nilaiBenar/jumlahSampah > 0.8){
    papanScore.bintang.gotoAndStop(3);
}else if (nilaiBenar/jumlahSampah > 0.4){
    papanScore.bintang.gotoAndStop(2);
}else{
    papanScore.bintang.gotoAndStop(1);
}
//tampilkan papanScore ke layar
addChild(papanScore);
}

setup();

```

8. Klik **Ctrl+Enter** untuk menjalankan program, jika berhasil maka akan tampil seperti gambar berikut ini:

